

# LA SORPRESA (DAAD ADVENTURE)



bienvenido a este tutorial sobre cómo desarrollar un juego usando el motor DAAD. En estas líneas podrás ir aprendiendo a usar este maravilloso motor para crear tu propia Aventura Gráfica y dejarla lista para jugar en un MSX. Este motor permite otros muchos sistemas de 8 bits, pero desconozco cómo funcionan, así pués, me he centrado en el sistema MSX que es el que conozco. ¡Que lo disfrutes!

## INTRODUCCIÓN


Esta es una aventura Survival Horror, que dicen los de habla inglesa, eso sí, con ciertos toques en clave de humor, de modo que la tensión suba y baje para no causar un ataque al corazón a ningún jugador.

Nuestro protagonista, Marcos, que vive en Madrid, recibe una llamada de su Padre para decirle que se reúna con la familia en su vieja cabaña en un bosque de la Sierra de Gredos. Tiene que contarle algo.

Todo el desarrollo de la historia ocurrirá en dicha cabaña y nuestro objetivo será lograr escapar de allí ileso. Ya entenderéis el por qué debemos salir de un sitio al que teníamos que entrar.

## MAPA

Este es el mapa en el que se desarrolla la primera parte del juego. Caja cajita es una **LOCalización**. Y esas líneas entre las cajas, son las **CONexiones** entre ellas, es decir, hacia dónde podemos dirigirnos/movernos desde dichas **LOCalizaciones**.



## COMIENZA LA AVENTURA

Lo primero que vamos a configurar son las **CONexiones** de las **LOCalizaciones**.

```
/CON ;Conexiones
/0
/1
E 1Coche
/2
E 1Extcasa
/3
N 1Lenera
S 1Garage
E 1Porche
/4
S 1Extcasa
/5
N 1Extcasa
```

/6
0 1Extcasa

Ahora debemos definir nuestras **LOCalidades** o **LOCalizaciones** que usaremos en el juego. Tiene que haber una por cada cajita de nuestro mapa y se ponen en el comienzo de la plantilla **DSF**, en la zona **/CTL**:

```
;localidades
#define lIntro 0
#define lMadrid 1
#define lCoche 2
#define lExtcasa 3
#define lLenera 4
#define lGarage 5
#define lPorche 6
```

## EXPLICACIÓN:

Lo primero de todo es decir que, por cada **CONexión**, puedes definir (`#define`) una **LOCalización** en **/CTL**. Si no, puedes referirte a ellas con su número (**AT 1**).

- **/0** es el comienzo de la aventura, la primera pantalla que aparece nada más cargar nuestro juego. En esta localidad no hay objetos ni acciones. Es sencillamente la presentación de nuestra aventura y una vez leída, pasa automáticamente a la siguiente localidad, la **/1**. El texto que se muestra aquí es el definido en la zona **/LTX** como **/0**. Y también debe existir una entrada de dicha localidad en **/CTL**, ya que dicha entrada o definición, será la que se use más adelante en las secciones de **PROcesos** o **/PRO** de nuestro fichero **DSF**. Ya, explicaremos esto de los **/PRO** más adelante.
- **/1** es la localidad siguiente después de mostrar la presentación de la aventura, como he explicado antes. Al igual que la anterior, mostrará el texto definido en la zona **/LTX** y la tendremos definida en **/CTL**. Aquí ya puede haber acciones, objetos, etc. Aquí comienza la interacción con el jugador que comienza a introducir comandos. Si os fijáis bien, en las **CONexiones**, tenemos para **/1** un **E 1Coche**, que sirve para indicar que desde la localidad **/1**, que es **lMadrid**, podemos movernos al **ESTE**, a la localidad **lCoche**.
- **/2** funciona de la misma manera que **/1** con la diferencia de que nos podemos mover la localidad **lExtCasa** si vamos al **ESTE**.
- **/3** sigue siendo igual con la diferencia de que tenemos tres posibles movimientos, **NORTE**, **SUR** y **ESTE** y nos desplazamos a **lLenera**, **lGarage** y **lPorche**.

No explico los demás, porque entiendo que ha quedado claro el funcionamiento del **/CON**.

Hemos comentado que cada vez que te mueves a una nueva localidad (desde ahora voy a llamarlo **LOCALización**, que me gusta más), mostramos un texto descriptivo de la misma. Pues bien, aquí es donde pondremos las descripciones de las **LOCALIZACIONES**, en la zona /LTX:

/LTX ;Location Texts

/0 "Es sábado, son las 08:00AM y aún sigues en la cama. Semi despierto y sin ganas de levantarte. Madrugas mucho durante la semana y no te apetece mucho abandonar el calorquito que sientes bajo el plumón.#n"

/1 "Te giras en la cama con la firme intención de dormir un poquito más. Suena el teléfono móvil. Alguien ya decidió por ti que te debes despertar del todo. Tienes a tu lado la mesita de noche."

/2 "Sales de casa y coges tu coche rumbo a la sierra de gredos desde Madrid. Es un paseito, pero el viaje se hace ameno escuchando un poco de música. Vete al Este"

/3 "Estás en el exterior de la casa. Desde aquí puedes ir al Norte, Sur y Este"

/4 "Estás en la leñera. Puedes entrar o volver al Sur"

/5 "Lograste entrar en el garage. Que fácil es todo cuando tienes las llaves"

/6 "Estás en el porche de la casa. Pero recuerdas que cuando tus padres salían a pasear por el bosque, solían esconderla en algún sitio. Lo recuerdas? Al Este está la puerta principal."

También vamos a definir algunos verbos nuevos que vamos a usar en la aventura y que no vienen por defecto en nuestra plantilla de inicio (**BLANK\_ES.DSF**). Se definen en la zona que pone **/VOC de VOCabulario**.

;verbos jugador  
  USAR 33 verb  
  DORMIR 34 verb  
  TIRAR 35 verb  
  ESCUCHAR 36 verb

Para esta primera parte, hemos definido el verbo **USAR** porque será lo que tendrá que hacer nuestro protagonista con el móvil, **USAR MOVIL**. De este modo, es como podrá continuar la aventura. Luego veremos el por qué.

En la parte principal, en la zona /CTL o **Sección de ConTroL**, vamos a añadir algunas cositas más.

Algunos **Flags** que necesitaremos para esta primera parte.  
Ahora explicaré para que los usaremos:

```
;Flags de usuario
#define fPuertaLenera 64
#define fPuertaGarage 65
#define fPuertaPorche 66
#define fUsarMovil 67
#define fNoUsarMovil 68
#define fExaminarMaceta 69
#define fCogerFelpudo 70
#define fCogerGnomo 71
```

Estos **Flags**, son los que vamos a usar durante la aventura para saber si abrió alguna puerta o si usó el móvil. Son condicionales para avanzar en la aventura, me explico, si por ejemplo no has usado el móvil, es decir, no decidiste coger la llamada de teléfono, no puedes seguir avanzando en la aventura.

Los **Flags** de las puertas, igual, si no se activan dichos **Flags**, no podrás acceder a nuevas **LOCALIZACIONES**, como la *Leñera*, el *Garage* o abrir la puerta principal de la casa ubicada en la localización *Porche de la casa*.

Estos **Flags** van a contener un 0 al inicio de la aventura y valdrán 1 cuando hayamos realizado ciertas acciones, como **COGER MOVIL** o **COGER LLAVE**. Aunque algunas llaves las vamos a coger de forma automática al descubrirlas.

Y ahora definimos todos los objetos que vamos a usar en estas **Localizaciones**.

```
;Objetos
#define oMovil 0
#define oLlave1 1
#define oLlave2 2
#define oLlavePorche 3
#define oelpudo 4
#define oMaceta 5
#define oGnomo 6
```

También tenemos que definir nuestros **OBJetos** en la zona **/OBJ**.

Este apartado es especialmente importante.

- **Obj:** es el número del objeto definido en /CTL
  - **Starts:** hay tres posibilidades:
    - 252: El objeto no está en ninguna parte (de momento no es visible).
    - 253: El objeto lo llevamos puesto (es una prenda).
    - 254: El objeto lo llevamos en nuestro inventario.
  - **Weight:** (Buscar documentación)
  - C: Indica si el objeto lo llevas en el Inventario (CARRIED)
  - W: Indica si el objeto lo llevas puesto (WEARED)
  - **Números:** (Buscar documentación)
  - **Noun:** Nombre del objeto
  - **Adjective:** (Buscar documentación)

## ACCIONES

Por cierto, no olvidéis que, por cada *nombre de objeto* que pongáis después del verbo, éstos deberán estar definidos en la zona /VOC.

```
; Objetos MESITA 99 noun  
MOVIL 100 noun  
LLAV1 101 noun  
LLAV2 102 noun  
LLAV0 103 noun  
FELPUDO 104 noun  
MACETA 105 noun  
GNOMO 106 noun  
PORCHE 107 noun  
METALICA 108 noun  
PUERTA 109 noun
```

Vamos a comenzar a introducir cierta funcionalidad a la aventura. Para eso vamos a hacer uso de la zona /PRO, en este caso /PRO 5.

```
> EXAMINA MESITA AT lMadrid  
MESSAGE "La mesita de noche está limpia y perfecta."  
ISAT oMovil 252  
MESSAGE "Hay un teléfono móvil en ella."  
CREATE oMovil  
DONE  
> EXAMINA MESITA AT lMadrid  
NOTCARR oMovil  
MESSAGE "El teléfono sigue sonando"  
DONE  
> EXAMINA MESITA AT lMadrid  
DONE
```

### EXPLICACIÓN DE LA PRIMERA ENTRADA:

La primera vez que pones EXAMINAR MESITA, vas a encontrar el objeto oMovil (lo hacemos visible en esta LOCalización) y lo creamos.

- **EXAMINA:** este es el verbo al que vamos a darle funcionalidad
- **MESITA:** es el objeto al cual podemos hacer uso del verbo
- **AT lMadrid:** indica que se puede usar en esta localización
- **MESSAGE:** El texto que mostramos al ejecutar esta acción.
- **ISAT oMovil 252:** Indicamos que hay un móvil. Si no ha usado antes EXAMINA MESITA, ese móvil no aparece por ningún sitio porque es un objeto oculto que deja de serlo al usar esta acción.
- **MESSAGE:** El texto que vamos a mostrar en pantalla tras descubrir el nuevo objeto.
- **CREATE oMovil:** Acabamos de crear el objeto y ya será siempre visible.
- **DONE:** Acaba la acción. No se ejecuta nada más.

### EXPLICACIÓN DE LA SEGUNDA ENTRADA: (SÓLO LAS DIFERENCIAS)

Si pones de nuevo EXAMINAR MESITA y NOTCARR oMovil, es decir, no hemos cogido el móvil aún (es el acrónimo de NOT CARRIED), el mensaje de texto "El teléfono sigue sonando" es una pista más que te indica que debes cogerlo.

- **NOTCARR oMovil:** Comprueba si aún no hemos cogido el móvil

- **MESSAGE:** El texto que mostramos
- **DONE:** Fin de acción

### EXPLICACIÓN DE LA ÚLTIMA ENTRADA:

Como ya hemos cogido el móvil, con la acción (COGER MOVIL), ahora sencillamente le mostramos un mensaje que describe la mesita.

## SEGUIMOS CON MÁS ACCIONES

Vamos a poner un par de acciones más para la LOCalización lMadrid:

```
> USAR MOVIL AT lMadrid  
CARRIED oMovil ; ¿lleva móvil?  
MESSAGE "+Buenos días hijo. Qué tal estás?.#n-Hasta que me llamaste  
estaba como un rey, en cama, calentito y tratando de negociar con  
Morpheo una segunda ronda.#n=Dile a Morpheo que espere un momento.  
Escucha, vente a la cabaña a pasar el día con nosotros, tenemos algo  
que contarte#n-Me apetece mucho, si... Pero no puede...#n+Hasta  
luego#n-Cojunto, no he podido poner un excusa, tengo que ir."  
SET fUsarMovil  
DONE  
> MIRAR _ AT lMadrid ; Si ya has cogido el móvil e insistes en mirar  
cosas  
CARRIED oMovil ; ¿lleva móvil?  
NOTZERO fUsarMovil  
MESSAGE "Aquí ya no tienes nada que hacer. Tira para el coche: Este"  
DONE  
> E _ AT lMadrid  
Zero fUsarMovil  
MESSAGE "No crees que deberías atender al móvil?"  
DONE
```

### EXPLICACIÓN DE LA PRIMERA ENTRADA:

Aquí vamos a mostrar la conversación con su padre, una vez cogido el móvil y lo use. Y además crearemos un Flag fUsarMovil para controlar que ha usado el móvil. Recordemos, si no usa el móvil, la aventura no puede avanzar.

- **USAR MOVIL:** la acción que vamos a controlar.
- **AT lMadrid:** en la localización de lMadrid
- **CARRIED oMovil:** controlamos que cogió el móvil.
- **MESSAGE:** Es el texto que aparece al coger el móvil. La conversación con su padre.
- **SET fUsarMovil:** Activamos el Flag de ha usado el móvil. El Flag se pone a 1.
- **DONE:** Acaba la acción.

### EXPLICACIÓN DE LA SEGUNDA ENTRADA:

Como se supone que ya hemos cogido el móvil y hablado con nuestro Padre, debemos ayudar al jugador a que dé el siguiente paso. Y para que no se ponga a enredar en la habitación porque NO hay nada más que hacer, le ayudamos con la acción MIRAR, que siempre que la usamos, nos muestra la descripción de dicha localización. Al definirla aquí, le podemos dar más pistas al jugador.

- **MIRAR \_ :** Es el verbo al que le definimos esta funcionalidad. Cuando use MIRAR lo que sea (ese \_ es lo que indica).
- **AT lMadrid:** en la localización lMadrid
- **CARRIED:** Comprobamos que lleve el móvil en su inventario
- **NOTZERO fUsarMovil:** Comprobamos que usó el móvil, es decir, que cogió la llamada.
- **MESSAGE:** El mensaje que mostramos al jugador.
- **DONE:** Acaba la acción.

## EXPLICACIÓN DE LA TERCERA ENTRADA:

Esta entrada es para controlar que no se vaya de la habitación a la **LOCalización 1Coche (Este)** sin haber cogido la llamada.

- **E \_**: Si quiere ir al este.
- **AT 1Madrid**: estando en dicha localización.
- **ZERO fUsarMovil**: comprobamos si el **Flag** es 0, es decir, que no ha usado el móvil, que no cogió la llamada.
- **MESSAGE**: El texto a mostrar.
- **DONE**: Acaba la acción.

## PRIMERA FORMA DE ACABAR LA PARTIDA

Vamos a introducir unas pocas líneas más en **/PRO 5** para inducir al jugador a seguir jugando, o de lo contrario, le sacamos de la partida. Sólo tiene una oportunidad, es decir, que si ejecuta **DORMIR** una vez, le damos un aviso, si lo hace de nuevo, se acabó la partida.

```
> DORMIR _ AT 1Madrid
NOTCARR oMovil ; No hemos cogido el móvil
ZERO fNoUsarMovil
SET fNoUsarMovil
MESSAGE "El móvil sigue sonando. No piensas cogerlo?"
DONE
> DORMIR _ AT 1Madrid
NOTZERO fNoUsarMovil
NOTCARR oMovil ; No hemos cogido el móvil
MESSAGE "Nada en este mundo va a romper tu momento de paz y
tranquilidad. Cuando te levantes, seguro que te hechas una partida al
nuevo juego de MSX, La Leyenda de Standard"
END
```

## EXPLICACIÓN DE LA PRIMERA ENTRADA:

Si el jugador, nos vacila y lo que quiere es **DORMIR**, le recordamos que el móvil sigue sonando, para dejarle claro que hay que cogerlo.

- **DORMIR \_**: la acción que vamos a controlar.
- **AT 1Madrid**: en la localización de **1Madrid**
- **NOTCARR oMovil**: controlamos que NO cogió el móvil.
- **ZERO fNoUsarMovil**: comprobamos si el Flag es 0
- **SET fNoUsarMovil**: activamos el Flag de que NO ha usado el móvil a 1. De este modo, si escribe de nuevo **DORMIR**, en la segunda entrada lo sacamos fuera de la partida.
- **MESSAGE**: Es el texto que aparece.
- **DONE**: Acaba la acción.

## EXPLICACIÓN DE LA SEGUNDA ENTRADA:

Si el jugador, nos vacila de nuevo, lo sacamos de la partida.

- **DORMIR \_**: la acción que vamos a controlar.
- **AT 1Madrid**: en la localización de **1Madrid**
- **NOTCARR oMovil**: controlamos que NO cogió el móvil.
- **NOTZERO fNoUsarMovil**: comprobamos si el **Flag** NO es 0. Será 1 porque ya puso **DORMIR** una primera vez.
- **MESSAGE**: Es el texto que aparece.
- **END**: Fin de partida.

## HEMOS SALIDO DE LA HABITACIÓN

Con el móvil en nuestro inventario, le dejamos salir de la habitación. Ya le informamos en su momento que, para salir de la habitación debía irse al Este.

Estas nuevas acciones son para controlar cuando salga, se meta en el *coche* y sin posibilidad de introducir otro comando, del *coche* le llevamos al *Exterior de la casa*.

```
> E _ AT 1Madrid ; Si no has usado el móvil y quieres salir al Este
NOTZERO fUsarMovil
CARRIED oMovil
MESSAGE "Cojonudo! Te has dejado el móvil en casa. No puedes volver
porque vas con prisas."
DESTROY oMovil
INKEY
CLS
GOTO 1Coche
RESTART
DONE
> E _ AT 1Coche
GOTO 1Extcasa
CLS
RESTART
DONE
```

## EXPLICACIÓN

- **DESTROY**: destruimos el objeto **oMovil**
- **INKEY**: espera la pulsación de una tecla
- **CLS**: limpiamos la pantalla (**CLearScreen**)
- **RESTART**: Cancelará cualquier bucle **DOALL**, cualquier llamada de subprocesso y realizará un salto para ejecutar el proceso 0 nuevamente desde el principio. (Sacado de la documentación oficial). Lo que significa es que cuando vayamos a una localización, cargue todo. La imagen si la tiene y su texto del **/LTX**

## YA ESTAMOS EN EL PORCHE DE LA CASA

Bien, tenemos dos **LOCalizaciones** que no hemos comentado porque no hay mucho que decir de ellas (de momento), por ejemplo, **1Coche**, que simplemente es descriptivo y le decimos al jugador que tire para el **Este**. La otra es la **1ExtCasa**, el *Exterior de la Casa*, que tampoco podemos hacer nada excepto ir al **Este** (**IPorche**).

Ahora bien... ¿realmente pensáis que hemos puesto dos localizaciones que no vamos a usar? NO. Lo cierto es que para esta primera parte del juego, estas dos **LOCalizaciones** son de paso, pero para el final del juego, ¡van a ser claves para poder acabarlo!

Centrémonos entonces en el *Porche de la casa*. Esta es la primera **LOCalización** dónde vamos a controlar un montón de posibles acciones para poder entrar en la casa. Vamos a controlar los objetos **MACETA**, **GNOMO** y **FELPUDO**. Y dejarle al jugador hacer algunas cosas con ellos.

Todas estas nuevas entradas van al apartado /PRO 5

```
> EXAMINAR MACETA AT lPorche
NOTCARR oMaceta
MESSAGE "Igual es mejor coger la maceta para verla más de cerca"
DONE
> EXAMINAR MACETA AT lPorche
CARRIED oMaceta
ZERO fExaminarMaceta
SET fExaminarMaceta
MESSAGE "La maceta contiene aún arena y una planta que está más seca
que la mojama."
DONE
> EXAMINAR MACETA AT lPorche
NOTZERO fExaminarMaceta
MESSAGE "Yo creo que ya está bien de jugar con la maceta."
DONE
```

#### EXPLICACIÓN DE LA PRIMERA ENTRADA:

Esta entrada es para cuando queremos echar un ojo a la maceta pero NO la hemos cogido.

- **EXAMINAR MACETA:** la acción que vamos a controlar.
- **AT lPorche:** en la localización de **lPorche**
- **NOTCARR oMaceta:** controlamos que NO cogió maceta
- **MESSAGE:** Es el texto que aparece.
- **DONE:** Acaba la acción.

#### EXPLICACIÓN DE LA SEGUNDA ENTRADA:

Esta entrada es para cuando queremos echar un ojo a la maceta pero SI la hemos cogido. Además, activamos unos Flags que se usarán en la tercera entrada.

- **EXAMINAR MACETA:** la acción que vamos a controlar.
- **AT lPorche:** en la localización de **lPorche**
- **CARRIED oMaceta:** controlamos que NO cogió maceta
- **ZERO fExaminarMaceta:** controlamos que NO cogió maceta
- **SET fExaminarMaceta:** controlamos que NO cogió maceta
- **MESSAGE:** Es el texto que aparece.
- **DONE:** Acaba la acción.

#### EXPLICACIÓN DE LA TERCERA ENTRADA:

Esta entrada es para cuando vuelve a ejecutar un examinar objeto y le dejamos claro, que ya no tiene nada que hacer con él.

- **EXAMINAR MACETA:** la acción que vamos a controlar.
- **AT lPorche:** en la localización de **lPorche**
- **NOTZERO fExaminarMaceta:** controlamos que NO cogió maceta
- **MESSAGE:** Es el texto que aparece.
- **DONE:** Acaba la acción.

Vamos a añadir algunas líneas para cuando el jugador, quiera examinar el FELPUDO y el GNOMO.

```
> EXAMINAR FELPUDO AT lPorche
MESSAGE "Es un clásico felpudo de esparto con el mensaje WELCOME
HOME."
DONE
> EXAMINAR GNOMO AT lPorche
NOTCARR oGnomo
ZERO fCogerGnomo
SET fCogerGnomo
ISAT oGnomo 252
CREATE oGnomo
```

```
MESSAGE "Es un gnomo rechoncho, con sonrisa picaresca y una mirada
intensa."
DONE
> EXAMINAR GNOMO AT lPorche
CARRIED oGnomo
NOTZERO fCogerGnomo
MESSAGE "Lo miras con atención y lo agitas en el aire. Suena algo
suelto dentro."
DONE
```

No voy a escribir las descripciones con detalle porque son exactamente igual que las anteriores pero para diferentes Objetos y un Flag que controla si cogió o no el Gnomo.

Ahora vamos a poner una entrada más muy interesante. Con esta acción, **MIRAR**, lo que hacemos es que sean visibles los objetos oMaceta, oFelpudo y oGnomo en la LOCalización **lPorche**. Además los creamos, de modo que ya se puede interactuar con ellos.

```
> MIRAR _ AT lPorche
ISAT oMaceta 252
CREATE oMaceta
ISAT oFelpudo 252
CREATE oFelpudo
ISAT oGnomo 252
CREATE oGnomo
MESSAGE "A tu alrededor puedes ver en el suelo una vieja maceta con
lo que debió ser una planta, un felpudo y un gnomo de jardín."
DONE
```

#### EXPLICACIÓN DE LA PRIMERA ENTRADA:

Esta entrada es para cuando queremos echar un ojo a todo lo que vemos desde la LOCalización **lPorche**.

- **MIRAR \_:** la acción que vamos a controlar.
- **AT lPorche:** en la localización de **lPorche**
- **ISAT oMaceta 252:** oMaceta ya es visible
- **CREATE oMaceta:** creamos el objeto
- **ISAT oFelpudo 252:** oFelpudo ya es visible
- **CREATE oFelpudo:** creamos el objeto
- **ISAT oGnomo 252:** oGnomo ya es visible
- **CREATE oGnomo:** creamos el objeto
- **MESSAGE:** Es el texto que aparece.
- **DONE:** Acaba la acción.

## AHORA LE DEJAMOS MANIPULAR ALGUNOS OBJETOS

Estas son algunas entradas para dejar a nuestro protagonista interactuar con los objetos: FELPUDO, MACETA y GNOMO

```
> COGER FELPUDO AT lPorche
ZERO fCogerFelpudo
SET fCogerFelpudo
MESSAGE "Debajo del felpudo no hay nada a excepción de un cienpies que sale por patas al haber sido descubierto."
DONE
> COGER FELPUDO AT lPorche
NOTZERO fCogerFelpudo
NOTCARR oLlavePorche
MESSAGE "La llave de la casa no va a aparecer sola. Sigue buscando."
DONE
> TIRAR MACETA AT lPorche
CARRIED oMaceta
MESSAGE "Tiras la maceta contra el suelo y no se rompe. De qué estarás hecha?"
DROP oMaceta
DONE
> DEJAR MACETA AT lPorche
CARRIED oMaceta
MESSAGE "Dejas la maceta en su sitio. Que buen hijo eres, si te viera tu madre."
DROP oMaceta
DONE
> DEJAR FELPUDO AT lPorche
MESSAGE "Dejas el felpudo en su sitio."
DONE
> TIRAR GNOMO AT lPorche
ISAT oLlavePorche 252
CREATE oLlavePorche
GET oLlavePorche
MESSAGE "Al tirar el gnomo contra el suelo y saltar en mil pedazos. Es una llave metálica. Me encanta ese grupo! Ahora bien, cuando se entere tu madre que has roto el gnomo... vas a flipar!!!."
DESTROY oGnomo
DONE
```

## POR FIN, ENTRAMOS EN LA CASA

Con esta entrada, permitiremos entrar en la casa a nuestro protagonista, gracias a que tiene el objeto **oLlavePorche**.

```
ABRIR PUERTA AT lPorche
CARRIED oLlavePorche
MESSAGE "Abres la puerta con la llave y notas que la luz que viste desde el exterior se acaba de apagar. Te preparan una sorpresa? De esas de: "hacemos que nos estamos y luego gritamos todos a la vez: Bienvenido Marcos!!". Odias las sorpresas."
goto lDescansillo
RESTART
DONE
```

## EXPLICACIÓN

Yo lo uso para que al irse a la nueva **LOCalización** **lDescansillo**, recarge el texto que tengo puesto en **/LTX** y cargue la imagen asignada a dicha **LOCalización**, en caso de existir.

## JUGANDO CON LA MESITA

Con estas entradas dejamos que el protagonista, examine la mesita y obtenga varios objetos. De los objetos que obtiene, el más importante son las cerillas. Porque este objeto, será clave en el juego. Cada vez que se use una cerilla, nuestro protagonista podrá moverse por **LOCalizaciones** oscuras pero esa cerilla, se apagará después de varios turnos, de modo que tiene que cuidar muy mucho cuándo usarlas y dónde.

```
> EXAMINAR MESITA AT lDescansillo
ZERO fMesita
SET fMesita
MESSAGE "Es una preciosa mesita de madera con patas forjadas de hierro. Tu madre la heredó de sus padres. Tiene un cajón semiabierto."
DONE
> EXAMINAR MESITA AT lDescansillo
NOTZERO fMesita
ZERO fCajon
MESSAGE "La mesita sigue siendo preciosa. Abres el cajón y seguimos?"
DONE
> EXAMINAR MESITA AT lDescansillo
NOTZERO fMesita
NOTZERO fCajon
MESSAGE "La mesita sigue siendo preciosa. Seguimos?"
DONE
> ABRIR CAJON AT lDescansillo
ISAT oLlave1 252
ISAT oLlave2 252
ISAT oCerillas 252
MESSAGE "En el cajón de la mesita puedes ver las llaves del garage y de la leñera, además de unas cerillas. Lo coges todo!!!"
CREATE oLlave1
GET oLlave1
CREATE oLlave2
GET oLlave2
CREATE oCerillas
GET oCerillas
ZERO fCerillas
SET fCerillas
LET fCerillas 15
ZERO fCajon
SET fCajon
MESSAGE "Tienes una caja con 15 cerillas."
DONE
> ABRIR CAJON AT lDescansillo
NOTZERO fCajon
MESSAGE "Ya no hay nada interesante en la mesita."
DONE
```

## CONTROL DE MOVILIDAD

Ahora, vamos a ver todos los controles de movilidad que ponemos a nuestro protagonista, es decir, desde las **LOCalizaciones** en las que se encuentre, dónde le dejamos desplazarse.

```
> BAJAR _ AT lEscaleras
CLEAR fDark
GOTO lDescansillo
RESTART
DONE
> E _ AT lEscaleras
ZERO fLuz
SET fDark
MESSAGE "No puedes moverte si vas a ciegas. Utiliza algo para iluminarte."
DONE
> E _ AT lPorche
MESSAGE "No puedes entrar en la casa. No tienes llaves."
DONE
> E _ AT lPorche
MESSAGE "No puedes entrar en la casa. No tienes llaves."
DONE
> E _ AT lPorche
CARRIED oLlavePorche
MESSAGE "Ya estás dentro de la casa. Hace frío. Y apenas hay luz. Te has tropezado con la mesita de la entrada.#n- Hola? Hay alguien en casa?#n-No quiero bromitas porque sino me voy a cabrear y lo sabéis!#nNo responde nadie."
goto lDescansillo
DONE
> E _ AT lMadrid ; Si no has usado el móvil y quieres salir al Este Zero fUsarMovil
MESSAGE "No crees que deberías atender al móvil?"
DONE
> E _ AT lMadrid ; Si no has usado el móvil y quieres salir al Este NOTZERO fUsarMovil
CARRIED oMovil
MESSAGE "Cojonudo! Te has dejado el móvil en casa. No puedes volver porque vas con prisas."
DESTROY oMovil
INKEY
CLS
GOTO lCoche
RESTART
DONE
> E _ AT lCoche
GOTO lExtcasa
CLS
RESTART
DONE
> N _ AT lExtcasa
Zero fPuertaLenera
MESSAGE "No puedes entrar en la leñera. La puerta está cerrada. Necesitas una llave."
DONE
> S _ AT lExtcasa
Zero fPuertaGarage
MESSAGE "No puedes entrar en el garaje. La puerta está cerrada. Necesitas una llave."
DONE
> E _ AT lPorche
Zero fPuertaPorche
MESSAGE "No puedes entrar la casa. La puerta está cerrada. Necesitas una llave."
DONE
```

